
ABSTRACT 
 
Managing and facilitating data integrity is a critical element when collecting information.  Many dollars are spent to 
actively obtain data, making it imperative to ensure the data is of the highest quality.  Even with interactive data entry 
and other checks performed at time of data entry, it will often fall on the SAS programmer to find the remaining 
problems and bring them to light.  The SAS system allows you to find, organize, and report data errors simply and 
effectively.  This paper, written from a clinical research perspective, presents a strategy for performing and facilitating 
data integrity checks using SAS, including how to communicate and resolve errors with study team members.  The 
examples shown use Base SAS® and work for versions 6 and above.  The programming presented is at a beginner’s 
level, but the strategy for managing data integrity applies to a broader audience. 
 
TERMS 
 
Data error – An incorrect variable value in a data set, which may include missing values. 
Data check – Logic placed on a variable or variables to identify data errors. 
Data cleaning – The process of finding data errors and fixing them. 
Study team – A group of people responsible for conducting a research study.  Members specialize in designing and 
conducting the study, collecting, entering, and analyzing data, and writing publications.  
 
INTRODUCTION 
 
Data is used to make or influence decisions that effect businesses and society.  Results of data collection include 
publishing manuscripts, papers, or articles in a journal, demonstrating the safety of a medical product, presenting at 
conferences or business meetings, and determining a business strategy.  People making decisions assume the 
results they reference are reliable.  Data integrity plays a large role in making sure that assumption is true.  Collecting 
and cleaning data for research studies often consumes the majority of project funds.  Given the volume of data 
collected and that it is often hand-entered from paper forms, it is not reasonable to expect that all data will be entered 
perfectly.  In fact, sometimes the data is entered correctly but the source document was filled out improperly.  Even 
when proactive measures are taken during data entry, such as screen control that will not allow entry of pre-specified 
erroneous values, errors can go undetected, especially those that involve consistency between multiple variables.  
The goal is to implement a scheduled SAS program early in the study that will succinctly summarize and clearly 
organize data errors, allowing the appropriate personnel to remedy the problems quickly.  The techniques for 
managing data integrity presented in this paper were developed while working on large clinical research studies.  
These same techniques apply to any situation where data errors may occur. 
 
IN THE BEGINNING 
 
The first step is designing good data collection forms.  This is a difficult and laborious process that will not be 
discussed here.  This paper will use sample data collection forms printed in Appendix A.  Data form PATIENTS 
collects patient demographics and data form FORM1 collects medical device information.  Next, someone, or some 
group, must actively go through the data collection forms, identifying the variables to check and how to check them.  
This phase will ideally include the SAS programmer.  In large studies it is often too time consuming to check all 
logical variable combinations.  Prioritization will help in this case.  Start by identifying checks that involve variables in 
your most important objectives or variables that will be difficult to fix after the study is completed.  These checks must 
then be translated into a SAS program. 
 
A SIMPLE START 
 
The core of the data check program is the SAS DATA step.  The data errors are found by writing IF-THEN-DO 
statements where the IF condition defines unacceptable data values.  Within the THEN-DO statements, a description 
of the error is put into a variable named ERROR, and the observation is output to the data set ERRORS.  It is 
important to note that only data errors are output as observations to ERRORS and that a single patient could 
generate multiple observations.  It is also a good idea to give output data sets names that identify the checked data, 

1 

Data Warehousing, Management and QualitySUGI 31

 



especially when checking data from multiple forms or combinations of forms within a single data check program.  Be 
advised that using the term ERROR in SAS program code will make searching the SAS log for processing and syntax 
errors more difficult.  The term ERROR is used in this paper for clarity, but another term is more practical in reality.  
Describing the errors meaningfully will greatly aid the cleaning process.  Know the intended audience and use error 
messages that make sense to them.  For example, avoid using SAS variable names in error messages when 
someone who is not a SAS programmer will use the report.  This technique may work when electronic data collection 
forms have SAS variable names annotated on them, allowing people to use an electronic ‘find’ utility, but in general 
does not perform well.   
 
Consider the following SAS statement appearing in a DATA step where the code in single quotes is an error message 
intended for a member of the study team: 
 
   Error = ‘var1 = 1 is not valid’ ; 
 
Seeing this message on a report does not provide enough detail to someone unfamiliar with SAS to find and fix the 
error.  This problem is solved by changing the error messages into a more useful form for those using the report and 
fixing the mistakes.  Rewrite the error messages using page numbers, question numbers, and text that reference the 
actual data collection forms, not the SAS data sets.  The sample data sets PATIENTS and FORM1 define simple 
demographic and medical device information for a group of 10 patients (Code Block 1).  These data sets contain data 
errors.  Code Block 2 presents a simple example showing how to check both a single variable’s value and for 
consistency across two variables.   
 
Code Block 1.  Sample data sets named PATIENTS and FORM1. 
 
data Patients ; 
    input PatientID Title $ Sex $ BirthDate mmddyy10. +1 DeathDate mmddyy10. ; 
    format BirthDate DeathDate mmddyy10. ; 
    datalines ; 
    1 Mr.  F 05/07/1945 07/01/1941 
   12 Mrs. F 01/23/1918 . 
    3 Mr.  M 12/03/1941 . 
    4 Miss F 07/06/1951 . 
    5 Ms.  F 10/09/1934 . 
    6 Mr.  M 04/04/1911 09/30/1984 
    7 Mr.  M 11/29/1926 . 
    8 Miss F 02/15/1949 . 
    9 Mr.  M 05/01/1948 . 
   10 Mrs. F 06/19/1939 05/10/1975 
    ; 
run ; 
 
 
data Form1 ; 
    input PatientID Implanted $ ImplantDate mmddyy10. DeviceType $ History     
          FluidStatus Shockless Total ; 
    format ImplantDate mmddyy10. ; 
    datalines ; 
    1 Yes 08/24/2004 ICD   1 1 1 4 
    2 No  .          .     . . . . 
    3 Yes 01/29/2001 IPG   . . . . 
    4 Yes 04/21/2005 ICD   . . . . 
    5 No  .          IPG   . . . . 
    6 Yes 11/03/1985 IPG   1 0 0 1 
    7 Yes 09/14/1996 ICD   1 1 . . 
    8 Yes 02/15/1949 IPG   . . . . 
    9 No  .          .     . . . 8 
   10 Yes 03/21/1974 ICD   2 1 1 3 
    ; 
run ; 

Code Block 2.  Code to identify data errors.  The variable ERROR describes the data error and the data form name 
is displayed in parentheses.  The form PATIENTS does not use question numbers, so they cannot be displayed. 
 
data Patient_Errors ; 
    set Patients ; 
    length Error $75 ; 
 
    /* Checking a single variable. */ 
    if PatientID > 10  
    then do ; 
        Error = 'ID is greater than 10, we only assigned IDs 1-10 (PATIENTS)' ; 
        output ; 
    end ; 
 
    /* Checking across two variables. */ 
    if (title = 'Mr.' and Sex = 'F') 
       or (title in ('Ms.', 'Mrs.', 'Miss') and sex = 'M') 
    then do ; 
        Error = 'Title and Sex indicate different genders (PATIENTS)' ; 
        output ; 
    end ; 
 
    /* Checking across two variables. */ 
    if . < DeathDate < BirthDate then do ; 
        Error = 'Death date precedes birth date (PATIENTS)' ; 
        output ; 
    end ; 
 
    keep PatientID Error ; 
run ; 
 
 
The resulting data set PATIENT_ERRORS contains three observations, meaning three data errors were identified.  
The next section will show how to report these errors. 
 
ORGANIZING AND REPORTING 
 
Only the SORT procedure and the PRINT procedure are needed for organizing and reporting the data check output.  
The FREQ procedure is used as an extra feature to create a summary showing the number of times each data error 
occurs (Code Block 3).  The PROC FREQ output is useful to highlight common mistakes and indicate possible logic 
flaws in the program.  For instance, if all patients have the same error, it is likely the IF-THEN-DO statements have 
the wrong logic.  The REPORT procedure could be used instead of PROC PRINT if desired.  Using PROC SORT 
prepares the data for PROC PRINT, allowing the use of the BY and ID statements with the PATIENTID variable.  The 
ID statement in PROC PRINT works nicely to visually separate different patients in the output file (Output 1).  The 
objective is to group the data by the PATIENTID variable so all errors from a given patient will be displayed together.  
Organizing the output this way allows for great efficiency when cleaning is performed.  To reconcile the errors, study 
team members must reference the original data source (e.g., CRF, electronic form, interview collected on a paper 
form) and the database.  This means they are working on a single patient at a time.  Listing errors by patient saves 
team members time because they can fix all problems for that patient and then move on to the next.  If the report was 
printed by the variable ERROR you would potentially have to revisit a patient multiple times as you worked through 
the list.  This may seem trivial for this example, but sometimes systems of data checks require multiple programs and 
data sets and are very complex.  In these cases the need to bring all the error observations into one data set and sort 
them by PATIENTID is critical. 
 
Code Block 3.  Code to sort, print, and summarize data errors. 
 
proc sort data = Patient_Errors ;     
    by PatientID ; 
run ; 
 
title3 'Detailed listing of patient data errors' ; 
proc print data = Patient_Errors ;     

    by PatientID ;     
    id PatientID ;     
    var Error ; 
run ; 
title3 ; 
 
title3 'Summary listing of patient data errors' ; 
proc freq data = Patient_Errors ;     
    table Error ; 
run ; 
title3 ; 
 
 
Output 1.  Listing output from Code Block 3.  The study team uses this output to reconcile the errors listed. 
 
                  Detailed listing of patient data errors 
 
Patient 
   ID                                 Error 
 
    1      Title and Sex indicate different genders (PATIENTS) 
           Death date precedes birth date (PATIENTS) 
 
   12      ID is greater than 10, we only assigned IDs 1-10 (PATIENTS) 
 
 
                                       Summary listing of patient data errors 
 
                                               The FREQ Procedure 
 
                                                                                      Cumulative 
Error                                                          Frequency     Percent   Frequency 
------------------------------------------------------------------------------------------------ 
Death date precedes birth date (PATIENTS)                             1       33.33           1 
ID is greater than 10, we only assigned IDs 1-10 (PATIENTS)           1       33.33           2 
Title and Sex indicate different genders (PATIENTS)                   1       33.33           3 
 
 
GOING ACROSS THE FORMS 
 
Many times other technology or systems can catch errors upstream from the SAS programmer.  Interactive data entry 
systems often recognize erroneous values as they are typed into electronic entry screens.  In these cases, the 
system will not accept the value and alert the individual entering the data.  This person can instantly recognize their 
mistake and enter the correct value.  Some research studies are fortunate to have other personnel besides a SAS 
programmer who implement data checks like those demonstrated above (Code Block 2).  These checks are removed 
from the SAS programmer’s plate!  However, these processes usually do not remove all data errors from the 
database, meaning a SAS programmer will have to implement data checks on individual variables.  Also, the data 
checks developed by other personnel or engaged by screen control methods often do not go across forms, and thus 
do not check the consistency between variables on different data collection forms, assuming each form goes into a 
separate data set.  Herein lay an opportunity for the SAS programmer to assemble the multiple forms (i.e., data sets) 
and further cleanse the data. 
 
The first step is to merge the data sets (Code Block 4).  There are 11 observations in data set ALL because the 
PATIENTID error in data set PATIENTS is not fixed yet. 
 
Code Block 4.  Merge the data sets PATIENTS and FORM1 together so you can program data checks using 
variables on different forms. 
 
proc sort data = Patients ; 
    by PatientID ; 
run ; 
 
proc sort data = Form1 ; 
    by PatientID ; 

run ; 
 
/* 11 observations because of the PatientID problem. */ 
data All ; 
    merge Patients Form1 ; 
    by PatientID ; 
run ; 
 
 
Next, develop new checks for the program that look at single variables on the form FORM1 and that compare 
variables across forms PATIENTS and FORM1 (Code Block 5).  First, add checks to compare the answers of filter 
questions to filtered questions.  Filter questions control whether or not subsequent questions are to be answered or 
left blank.  In this case, if an ICD is not implanted, then the question detailing ICD features should not be asked, and 
the data set should contain missing values for the variables corresponding to that question (Appendix A, FORM1, 
Questions 1b and 1c).  Alternatively, if an ICD is implanted, the question detailing ICD features should be asked, and 
the data set should contain non-missing values for the variables corresponding to that question.  Second, add a 
check to verify that individual elements sum to the total.  The sample data collection form FORM1 allows someone to 
circle up to three possible features and then enter the total number of features circled (Appendix A, FORM1, Question 
1c).  Finally, add checks that involve variables from both data sets, such as identifying patients whose birth date, 
implant date, and death date fall in the incorrect order and patients implanted before age 30 because this is unusual.  
These patients need their dates reviewed since they may have an incorrect birth date or implant date.  Remember to 
write error descriptions that will be most useful for the people using the report.  
 
Code Block 5.  DATA step code to generate data errors.  The variable ERROR describes the data error, the  
data form name is displayed in parentheses, and the question numbers are displayed when available.   
 
data Merged_Errors ; 
    set All ; 
    length Error $80 ; 
 
    if PatientID > 10  
    then do ; 
        Error = 'ID is greater than 10, we only assigned IDs 1-10 (PATIENTS)' ; 
        output ; 
    end ; 
 
    if (title = 'Mr.' and Sex = 'F') 
       or (title in ('Ms.', 'Mrs.', 'Miss') and sex = 'M') 
    then do ; 
        Error = 'Title and Sex indicate different genders (PATIENTS)' ; 
        output ; 
    end ; 
 
    if . < DeathDate < BirthDate then do ; 
        Error = 'Death date precedes birth date (PATIENTS)' ; 
        output ; 
    end ; 
 
    /* Filter question checks using variables from Form1 */ 
    if Implanted = 'Yes' and DeviceType = '' then do ; 
        Error = 'Patient was implanted so device type should not be missing  
                 (FORM1 - QN 1 & 1b)' ; 
        output ; 
    end ; 
 
    if Implanted = 'No' and DeviceType ^= '' then do ; 
        Error = 'Patient was not implanted so device type should be missing  
                 (FORM1 - QN 1 & 1b)' ; 
        output ; 
    end ; 
 
    /* Total check */ 

    if sum(History, FluidStatus, Shockless) ^= Total then do ; 
        Error = 'Sum of features does not equal Total (FORM1 - QN 1c)' ; 
        output ; 
    end ; 
 
    /* Add checks using variables from Patients and Form1 */ 
    if . < DeathDate < ImplantDate then do ; 
        Error = 'Implant date (FORM1 - QN 1a) is after death date (PATIENTS)' ; 
        output ; 
    end ; 
 
    if . < ImplantDate < BirthDate then do ; 
        Error = 'Implant date (FORM1 - QN 1a) is before birth date (PATIENTS)' ; 
        output ; 
    end ; 
 
    if ImplantDate ^=. and BirthDate ^=.  
        and ((ImplantDate - BirthDate) / 365.25) < 30 then do ; 
        Error = 'Implanted before age 30. Check implant (FORM1 - QN 1a) and  
                 birth date (PATIENTS).' ; 
        output ; 
    end ; 
 
    keep PatientID Error ; 
run ; 
 
 
Now organize and report the data errors using PROC SORT, PROC PRINT, and PROC FREQ (Code Block 6). 
 
Code Block 6.  Code to sort, print, and summarize data errors. 
 
proc sort data = Merged_Errors ;     
    by PatientID ; 
run ; 
 
title3 'Detailed listing of all data errors' ; 
proc print data = Merged_Errors ;     
    by PatientID ;     
    id PatientID ;     
    var Error ; 
run ; 
title3 ; 
 
title3 'Summary listing of all data errors' ; 
proc freq data = Merged_Errors ;     
    table Error / nocum ; 
run ;  
title3 ; 
 
 
Finally view the output (Output 2). 
 
Output 2.  Listing output from Code Block 6.  The study team uses the output to reconcile the errors listed. 
                           
                            Detailed listing of all data errors 
 
Patient 
   ID      Error 
 
    1      Title and Sex indicate different genders (PATIENT) 
           Death date precedes birth date (PATIENT) 
           Sum of features does not equal Total (FORM1 - QN 1c) 
           Implant date (FORM1 - QN 1a) is after death date (PATIENT) 

    5      Patient was not implanted so device type should be missing (FORM1 - QN 1 & 1b) 
 
    6      Implant date (FORM1 - QN 1a) is after death date (PATIENT) 
 
    7      Sum of features does not equal Total (FORM1 - QN 1c) 
 
    8      Implanted before age 30. Check implant (FORM1 - QN 1a) and birth date (PATIENT). 
 
    9      Sum of features does not equal Total (FORM1 - QN 1c) 
 
   10      Sum of features does not equal Total (FORM1 - QN 1c) 
 

12      ID is greater than 10, we only assigned IDs 1-10 (PATIENT) 
 
 
                                   Summary listing of all data errors 
 
                                           The FREQ Procedure 
 
Error                                                                       Frequency  Percent 
------------------------------------------------------------------------------------------------ 
Death date precedes birth date (PATIENT)                                           1     9.09 
ID is greater than 10, we only assigned IDs 1-10 (PATIENT)                         1     9.09 
Implant date (FORM1 - QN 1a) is after death date (PATIENT)                         2    18.18 
Implanted before age 30. Check implant (FORM1 - QN 1a) and birth date (PATIENT).   1     9.09 
Patient was not implanted so device type should be missing (FORM1 - QN 1 & 1b)     1     9.09 
Sum of features does not equal Total (FORM1 - QN 1c)                               4    36.36 
Title and Sex indicate different genders (PATIENT)                                 1     9.09 
 
 
This report shows, among other things, that the patient with PATIENTID=1 has four errors to investigate and that the 
error “Sum of features does not equal Total” appears most often (Output 2).   
 
STRATEGY FOR MANAGING DATA CLEANING 
 
It is important to set up a systematic and proactive approach to identify and report data errors.  Strive to implement 
the data check program when data collection begins so there is less time for the output to become unmanageable.  
Starting early also allows you to monitor the effectiveness of the program, then add or delete checks as necessary 
throughout the lifetime of the data collection phase.  Part of managing data cleaning efforts involves teaching people 
to understand the SAS report and how to fix the errors.  It is important to show them the logic works both ways.  For 
example, when the error message reads, “death date precedes birth date”, it implies either death date or birth date 
could be incorrect, and both may need to be checked.  Continue to be available for questions and monitor the 
situation.  This is a team effort requiring the skills of multiple people.  Another important aspect of managing data 
cleaning is to make fixing data errors part of the study team’s weekly tasks.  Schedule the data check program to run 
periodically, for example, every Monday morning.  Distribute the output to the personnel responsible for fixing the 
data errors by printing it in their offices, e-mailing it as an attachment, or saving it on a network drive and e-mailing 
the location.  These people should spend a few hours every week resolving data errors.  It is even possible to allow 
them to run the data check program from their computers whenever they want without involving a SAS programmer 
using an executable file in UNIX, SAS/IntrNet® software, or the SAS® BI Server.  This proves handy when they want 
to see if the corrections they made properly fixed the errors or possibly introduced new errors.  Also, in cases where 
the same person enters data and fixes errors, they may recognize common mistakes and avoid them in the future.  
This strategy will help make the task of detecting and cleaning data errors less overwhelming and very proactive.  
Imagine running the data cleaning program a week before the study ends and having 2,000 data checks to sort 
through knowing your study team personnel will be assigned to other projects one week later!  It is best to engage in 
data cleaning early and often in order to keep your data as accurate as possible at all times.  Other periodic reports or 
interim analyses will benefit greatly from this practice.   
 
PROCESS IMPROVEMENTS 
 
Once the basic program is completed and operational, begin thinking about process improvements by obtaining 
feedback from users and asking, “How can I make this better?”  One problem you may encounter relates to the error 
description messages.  Using SAS variable names can prove difficult for the study team to understand.  From their 
view, messages referencing the data collection forms often make more sense.  Rewrite the error descriptions using 
the form name, page number, and question number whenever possible to solve this problem.  You may also put the 
form name, page number, and question number into their own variables instead of embedding them into the error 

7 

Data Warehousing, Management and QualitySUGI 31

 



description.  Try tailoring the output by the name of the person who collected the data.  This allows each person to fix 
the data they personally collected.  Two benefits are realized: division of workload and intimate knowledge of the data 
by the person correcting it.  Checking for missing data forms is a welcome enhancement.  For example, the person 
with PATIENTID=1 may not have an observation in data set FORM1, but they should since they have an observation 
in data set PATIENTS.  Adding some more SAS code to the data-cleaning program can easily locate these mistakes.  
Another improvement is identifying duplicates for the study team to reconcile (Peterson, 2006).  One final idea uses 
PUT= to display the actual variable values of the data error in question.  A solid process for keeping data clean is 
achieved by starting simple and building upon that foundation through learning from your experiences, the 
experiences of others and user feedback.   
 
CONCLUSION 
 
A lot of money, time, and effort are invested in data collection.  Taking a proactive approach to ensure the data is 
accurate will only enhance the investment.  More importantly, because people use reports and publications based on 
the data to make important business and societal decisions, it is critical that the data is sound.  SAS software can 
easily help facilitate the process of identifying and cleaning data errors by simply using the DATA step, PROC SORT, 
and PROC PRINT.  A well-implemented data cleaning process requires communication amongst team members, 
planning, and getting started early.  By following the strategy presented you can rest assured clean data will be 
waiting for you upon study conclusion.   

APPENDIX A.  Sample data collection forms. 
 
EXAMPLE DATA COLLECTION FORM PATIENTS 
 
Patient information. 
 
Patient ID: pre-filled  
 
_______        ______ 
Title              Sex 
 
____________     ____________    
Birth date             Death date 
 
 
 
EXAMPLE DATA COLLECTION FORM 1 
 
Survey of features used for patients implanted with implantable cardioverter defibrillators (ICD). 
 
Patient ID: pre-filled 
 
1. Does the patient have a cardiac medical device implanted? 

Yes     No 
 

⇒ If Yes:   
 
a. Date of implant: ______________________ 
 
b. What type of device was implanted, a pacemaker or a defibrillator? 
     IPG     ICD 
 
⇒ If ICD: 
 
c. What features does your defibrillator contain? 
    (circle all that apply) 
     History       Fluid Status       Shockless Rx 

 
Total number of features ___ 

